                       ESEMPIO DI KB in CLIPS


(defmodule MAIN (export deftemplate ?ALL))

(deftemplate goal (slot source) (slot dest) (slot client))

(deftemplate flight (slot s) (slot d) (slot comp) (slot flight#))

(deftemplate reservation (slot name) (slot s) (slot d) (slot comp) 

                         (slot flight#))

(deffacts flight-info

       (flight (s Torino) (d Roma) (comp AZ) (flight# AZ200))

       (flight (s Torino) (d Paris ) (comp AF) (flight# AF300))

       (flight (s Torino) (d Paris ) (comp AZ) (flight# AZ301))

       (flight (s Torino) (d Amsterdam ) (comp KL) (flight# KL003))

       (flight (s Torino) (d Frankfurt) (comp LH) (flight#  LH405))

       (flight (s Torino) (d Frankfurt) (comp LH) (flight#  LH406))

       (flight (s Frankfurt) (d NY) (comp LH) (flight#  LH450))

       (flight (s NY) (d Huston) (comp LH) (flight# LH451))

       (flight (s Frankfurt) (d San-Francisco) (comp LH) (flight#  LH452))

       (flight (s Paris) (d Boston ) (comp AF) (flight# AF319)) 

       (flight (s Paris) (d Boston) (comp AA) (flight# AA200))

       (flight (s Paris) (d San-Francisco) (comp AF) (flight#  AF319))

       (flight (s Amsterdam ) (d Boston) (comp KL) (flight#  KL900))

       (flight (s Roma) (d Osaka) (comp AZ) (flight# AZ305))

       (flight (s Roma ) (d Tokyo) (comp Japan) (flight# JL100))

       (flight (s Paris) (d Tokyo ) (comp AF) (flight# AF379)) 

       (flight (s Paris) (d Osaka) (comp Japan) (flight# JL101)))

(defrule simple-trip

        (goal (source ?s) (dest ?x) (client ?t))

         (flight (s ?s) (d ?x) (comp ?z) (flight#  ?w))

         =>

         (assert (reservation (name ?t) (s ?s) (d ?x) (comp ?z) (flight#  ?w)))

         (assert (done)))

(defrule int1

        (goal (dest ?x&San-Francisco|Boston|Huston))

        => (assert  (dest  USA))  (focus INC))

(defrule int2

        (goal (dest ?x&Osaka|Tokyo))

         => (assert  (dest  Japan))  (focus INC))

(defrule stop

(done) => (halt))

(defrule connect

        (goal (source Torino) (client ?t))

        (reservation (name ?t) (s ?s&~Torino))

        (flight (s Torino) (d ?s) (comp ?z) (flight#  ?w)) 

     => (assert (reservation (name ?t) (s Torino) (d ?s) (comp ?z) (flight#  ?w)))

        (assert (done)))

(defmodule INC (import MAIN ?ALL))

(defrule inc1

        (dest USA) =>

        (assert (stop Frankfurt) (stop Zurich) (stop Paris) (stop Amsterdam)))

(defrule inc2

        (dest Japan) =>

        (assert (stop Roma) (stop Paris)))

(defrule inc3

        (goal (dest ?d) (client ?t))

        (stop ?x)

        (flight (s ?x) (d ?d) (comp ?z) (flight# ?w))

         =>

        (assert (reservation (name ?t) (s ?x) (d ?d) (comp ?z) (flight# ?w)))

        (pop-focus))

(assert (goal (source Torino) (dest Boston) (client PT)))

(assert (goal (source Torino) (dest Huston) (client PT)))

(assert (goal (source Torino) (dest San-Francisco) (client PT)))

